

网格交易

- 投资者设定某个证券的价值中枢，利用“档位”的模式对投资标的的进行机械式操作。下跌时，进行分批分档买入，上涨时，进行分批分档卖出。网格交易法由于不依赖人为的思考，完全是一种程序行为，像渔网一样，利用行情的波动在网格区间内低买高卖，避免追涨杀跌，拥有较强的操作纪律。下列是利用大智慧公式语言策略交易平台实现网格交易策略的例子。
- **本网格交易实现方法是同时发送两笔委托，一笔买入、一笔卖出，无股票底仓不会发单。**

策略学习案例仅供学习使用，用户不得直接用于实盘交易。

```
{  
  
    程序思路：  
  
    1、定义初始变量和常用参数  
  
    2、检查电脑时间在交易时间范围之内  
  
    3、计算基准价格，计算网格幅度  
  
    4、下两笔委托单  
  
    5、其中成交后，撤销另一笔，重新下两笔委托单。  
  
}  
  
{网格交易：策略开始时间、策略结束时间、网格大小(元)、每格买入数量、每格卖出数量、持仓上限、  
持仓下限、价格上限、价格下限}  
  
{定义变量}  
  
VARIABLE: 提示信息='', 起步基准价=0.0, 最新基准价=0.0;  
  
VARIABLE: 新网格大小=0.0;  
  
VARIABLE: 策略证券代码 = '', STOCKINFO1='';  
  
VARIABLE: 涨停价=0.0, 跌停价=0.0, 最新价=0.0;  
  
VARIABLE: 可用资金=0.0, 买入需要资金=0.0, BUY_PRICE=0.0, SELL_PRICE=0.0;  
  
VARIABLE: 下两笔网格单='', 买委托号='', 卖委托号='', 买委托状态 = '', 卖委托状态 = '';  
  
VARIABLE: 撤单失败委托号='';  
  
VARIABLE: BUY_ORDERINFO = '', SELL_ORDERINFO='', RETCD='';
```

```

VARIABLE:WTH1='', WTH2='', 资金持仓就绪=0;

输出日志 := 1;

开始时间 := 93000;

结束时间 := 113000;

开始时间 2 := 130000;

结束时间 2 := 235900;

网格买入数量 := 100;

网格卖出数量 := 100;

数量上限 := 1000000;    {数量上限, 达到上限则不再买入}

数量下限 := 0;        {数量下限, 达到下限则不再卖出}

价格上限 := 1030.00;   {价格上限, 达到上限则不再买卖}

价格下限 := 1.00;     {价格下限, 达到下限则不再买卖}

网格幅度 := 0.5;      {%}

网格类型 := 'YUAN';    {元 YUAN 或幅度 FD}

IF PRECISION >2 THEN

    网格大小 := 0.003;    {元}

ELSE

    网格大小 := 0.03;     {元}

{连续交易时间}

策略证券代码 := MARKETLABEL+STKLABEL;

IF TGetMarketStatus() = 3 THEN

    提示信息:='当前时间在连续交易时间范围内';

ELSE

    提示信息:='当前时间不在连续交易时间范围内';

```

```
{TWriteLog(输出日志,提示信息);}

本地电脑时间 := CURRENTTIME;

在时间范围内 := (本地电脑时间 >= 开始时间 AND 本地电脑时间 <= 结束时间) OR (本地电脑时间 >=
开始时间 2 AND 本地电脑时间 <= 结束时间 2);

IF 在时间范围内 > 0 THEN

    提示信息:='当前时间在策略运行时间范围内';

ELSE

    提示信息:='当前时间不在运行时间范围内';

可用资金 := 0;

可卖数量 := 0;

持仓数量 := 0;

涨停价 := DYNAINFO(54);

跌停价 := DYNAINFO(55);

最新价 := DYNAINFO(7);

IF 最新价 > 价格下限 AND 最新价 < 价格上限 THEN

    提示信息 := '最新价在价格上下限范围内';

ELSE

    提示信息 := '最新价不在价格上下限范围内';

IF 在时间范围内 > 0 AND 最新价 > 价格下限 AND 最新价 < 价格上限 THEN

BEGIN

    {保存策略开始时的最新价作为网格起步基准价格}

    IF 起步基准价 < 0.001 AND 最新价 > 0 THEN BEGIN

        起步基准价格 := 最新价;
```

```

IF 最新基准价 < 0.001 THEN

    最新基准价:=最新价;

IF STRCMP(网格类型,'FD') = 0 THEN BEGIN

    IF PRECISION >2 THEN

        新网格大小 := (FLOOR(最新价 * 网格幅度*1000)/1000)*1000;

    ELSE

        新网格大小 := (FLOOR(最新价 * 网格幅度*100)/100)*100;

    END

ELSE

    新网格大小 := 网格大小;

IF 新网格大小 < 0.001 AND PRECISION > 2 THEN 新网格大小 := 0.002;

IF 新网格大小 < 0.01 AND PRECISION <= 2 THEN 新网格大小 := 0.02;

END

{-----取资金信息-----}

可用资金 := TGetAvailCapital();

{-----取持仓信息-----}

策略证券代码 := MARKETLABEL+STKLABEL;

STOCKINFO1 := AGetHoldInfo();

IF STRCMP(STOCKINFO1,'ERROR') <> 0 THEN 持仓数量 :=
STRTONUM(STRSPLIT(STOCKINFO1,'|',3));

可卖数量 := TGetHoldAvailAmount();

END

下两笔网格单 := '';

```

```

IF 持仓数量 >= 数量下限 AND 持仓数量 <= 数量上限 AND 最新基准价 > 0.001 THEN BEGIN

    买委托状态 := '';

    卖委托状态 := '';

    IF 新网格大小 > 0.001 THEN BEGIN

        网格大小 := 新网格大小;

        {检查买入委托状态、卖出委托状态}

        IF STRCMP(买委托号, '') <> 0 THEN BEGIN

            {0: 失败; 1: 未报; 2: 已报; 3: 部分成交; 4: 全部成交; 5: 作废; 6: 全部撤单; 7:
            部成部撤; 8: 撤单已报。}

            IF TGetOrderStatus(买委托号) = 4 THEN

                BEGIN

                    提示信息 := '买委托号:' + 买委托号 + '已经成交';

                    TWriteLog(输出日志, 提示信息);

                    最新基准价 := 最新基准价 - 网格大小;

                    IF STRCMP(卖委托号, '') <> 0 THEN BEGIN

                        RETCD := ACancelOrder(卖委托号);

                        IF STRCMP(RETCD, 'OK') = 0 THEN BEGIN

                            提示信息 := '撤销卖委托号:' + 卖委托号 + ',成功';

                            TWriteLog(输出日志, 提示信息);

                            买委托号 := '';

                            卖委托号 := '';

                        END

                    ELSE BEGIN

                        提示信息 := '撤销卖委托号:' + 卖委托号 + ',失败';

                        TWriteLog(输出日志, 提示信息);

```

```

        撤单失败委托号 := 卖委托号;

        买委托号 := '';

        卖委托号 := '';

    END

END

END

ELSE

    提示信息 := '买委托状态='+买委托状态;

END

ELSE

    提示信息 := '买委托号为空';

IF STRCMP(卖委托号, '') <> 0 THEN BEGIN

    {0: 失败; 1: 未报; 2: 已报; 3: 部分成交; 4: 全部成交; 5: 作废; 6: 全部撤单; 7:
    部成部撤; 8: 撤单已报。}

    IF TGetOrderStatus(卖委托号) = 4 THEN

        BEGIN

            提示信息 := '卖委托号:'+卖委托号+'已经成交';

            TWriteLog(输出日志, 提示信息);

            最新基准价:=最新基准价 + 网格大小;

            IF STRCMP(买委托号, '') <> 0 THEN BEGIN

                RETCD := ACancelOrder(买委托号);

                IF STRCMP(RETCD, 'OK') = 0 THEN BEGIN

                    提示信息:='撤销买委托号:'+买委托号+',成功';

                    TWriteLog(输出日志, 提示信息);

                    买委托号 := '';

```

```
        卖委托号 := '';

    END

    ELSE BEGIN

        提示信息 := '撤销买委托号:' + 买委托号 + ',失败';

        TWriteLog(输出日志, 提示信息);

        撤单失败委托号 := 买委托号;

        买委托号 := '';

        卖委托号 := '';

    END

    END

    END

    ELSE

        提示信息 := '卖委托状态=' + 卖委托状态;

    END

    ELSE

        提示信息 := '卖委托号为空';

    IF STRCMP(买委托号, '') = 0 AND STRCMP(卖委托号, '') = 0 THEN BEGIN

        下两笔网格单 := 'TRUE';

        提示信息 := '下两笔网格单=' + 下两笔网格单;

    END

    ELSE

        提示信息 := '买委托号=' + 买委托号 + ',卖委托号=' + 卖委托号;

    IF STRCMP(下两笔网格单, 'TRUE') = 0 THEN BEGIN
```

```

下两笔网格单 := '';

BUY_PRICE := 最新基准价 - 网格大小 ;

SELL_PRICE:= 最新基准价 + 网格大小 ;

买入需要资金 := BUY_PRICE * 网格买入数量 + (BUY_PRICE * 网格买入数量)*5/10000;

资金持仓就绪 := (买入需要资金 <= 可用资金) AND (网格卖出数量 <= 可卖数量);

IF 买入需要资金 > 可用资金 THEN BEGIN

    提示信息 := 策略证券代码+', 买入需要资金='+NUMTOSTR(买入需要资金)+'>可用资金='+NUMTOSTR(可用资金);

    TWriteLog(输出日志, 提示信息);

END

IF 网格卖出数量 > 可卖数量 THEN BEGIN

    提示信息 := 策略证券代码+', 网格卖出数量'+NUMTOSTR(网格卖出数量)+'>可卖数量='+NUMTOSTR(可卖数量);

    TWriteLog(输出日志, 提示信息);

END

{判断最新基准基准价格有效性}

IF 资金持仓就绪 > 0 AND 最新基准价 - 网格大小 >= 跌停价 AND 最新基准价 + 网格大小 <= 涨停价 THEN

BEGIN

    WTH1 := ASendOrder(1,1,BUY_PRICE, 网格买入数量);

    IF STRCMP(WTH1, 'ERROR') = 0 THEN BEGIN

        提示信息 := '买入:'+策略证券代码+', 价格:'+NUMTOSTR(BUY_PRICE)+' , 数量:'+NUMTOSTR(网格买入数量)+' , 失败';

        TWriteLog(输出日志, 提示信息);

        买委托号 := '';

```

```

END

ELSE BEGIN

    买委托号 := WTH1;

    提示信息 := '买入:'+策略证券代码+', 价格:'+NUMTOSTR(BUY_PRICE)+' ,数
量:'+NUMTOSTR(网格买入数量)+' ,委托号:'+WTH1;

    TWriteLog(输出日志, 提示信息);

    WTH2 := ASendOrder(2,1,SELL_PRICE, 网格卖出数量);

    IF STRCMP(WTH2, 'ERROR') = 0 THEN BEGIN

        提示信息 := '卖出:'+策略证券代码+', 价格:'+NUMTOSTR(SELL_PRICE)+' ,
数量:'+NUMTOSTR(网格卖出数量)+' ,失败';

        TWriteLog(输出日志, 提示信息);

        卖委托号 := '';

        RETCD := ACancelOrder(买委托号);

        IF STRCMP(RETCD, 'OK') = 0 THEN

            提示信息:='撤销买委托号:'+买委托号+', 成功';

        ELSE BEGIN

            提示信息:='撤销买委托号:'+买委托号+', 失败';

            撤单失败委托号 := 买委托号;

        END

        TWriteLog(输出日志, 提示信息);

        买委托号 := '';

        卖委托号 := '';

    END

ELSE BEGIN

```

```
        卖委托号 := WTH2;

        提示信息 := '卖出:'+策略证券代码+',价格:'+NUMTOSTR(SELL_PRICE)+' ,
数量:'+NUMTOSTR(网格卖出数量)+' ,委托号:'+WTH2;

        TWriteLog(输出日志, 提示信息);

    END

END

END

END

END

ELSE

    提示信息 := '下两笔网格单<>TRUE';

END

1;
```